教材としての人口推定モデル

植　竹　恒　男

1. はじめに

昨年の9月17日から23日まで西ドイツの Bielefeld で JOINT ICMI/ICPE/CTS/UNESCO/IDM-CONFERENCE ON COOPERATION BETWEEN SCIENCE TEACHERS AND MATHEMATICS TEACHERS という長い名前の会議が開かれた。

最初の三つの略号はそれぞれ International Commission on Mathematical Instruction, International Committee on Physics Education, Committee on Teaching of Science という国際的な組織を意味し，最後の IDM は Institute for the Didactics of Mathematics という Bielefeld 大学内の機関を意味している。

この会議に，わが国からは木下是雄氏（物理学，学習院大学）が出席された。

昨年12月26，27日に大磯で行なわれた「他科学と総合した数学カリキュラムに関する研究（科学研究費補助金による)」の総会でその報告があった。筆者は，この研究班の一人として，早くから木下氏のもたらされた資料を入手し，その一部を分担，検討して，上記の会合で発表した。以下の小文は，そのときの発表と，その後検討した結果とをまとめたものである。

2. 数学と他科学との関連について

表記については，古代ギリシャにおけるユークリッドの〈原論〉と諸科学との関連，および，17世紀における，微分積分学と物理学（特に力学）との関連がよく知られている。

前者の結びつきは，西欧における合理主義の根幹となり，後者による所産は，こんにちの科学技術の基礎を形づくっている。学校教育における「数学」の内容も，これらのことを念頭において選ばれてきた。

さらに20世紀に入って，数学は，いままで考えられなかったようないろいろな科学の分野と関連をもつようになってきた。このようにして生まれた境界領域は「数理科学」とよばれる。ここで用いられる主な数学としては，微分積分学のほかに，推測統計学や，線形代数学がある。この新しい状況に対応するために学校教育の内容も改められていかなければならない。この種の研究は今なお進行中であり前項で挙げた研究組織もそのために作られたものである。

数理科学における方法は，大まかにいって

$$\text{現　象} \rightleftarrows \text{数学モデル}$$

という形で示される。数学モデル（Mathematical Model）は，現象の特徴を数学的に表現したものであり，それを数学的に解いた結果の解釈と，実際の現象へのフィードバックが矢印←—である。その結果，モデルは手直しされ，また，同じようなプロセスがくりかえされるかもしれない〔文献(1)〜(4)，(7)，(8)〕。学校教育においても，簡単な数学モデルを用いて，このようなプロセスを理解させていかなければならないであろう。

前項で述べた CONFERENCE においても，この問題は，第1の working-group で

Science Applications and Model Building as Part of Mathematical Curricula

という形でとり上げられ，それに適した数学モデルを集めた書物を作ることが提案された由である。文献(5)，(6)はその際に配布された papers であって，これを各国に持ち帰って検討することが期待されている。

3. 人口推定モデル（その1）

文献(5)には，6種の数学モデルが集められているが，ここでは，その一つ，「人口推定モデル」をとり上げよう。

まず，一番単純なモデルから始める。ある年の始めの人口を x_0，その後，各年始めの人口を x_1，x_2，……とすれば，1年間の人口の変化は $x_{i+1}-x_i$ で表される。ただし，$i=0, 1, 2,$ ……とする。

この値は，出入国する人数を考えに入れなければ，その1年間の出生数から，死亡数を引いた差に等しい。そこで，これらの値が，いずれも x_i に比例すると仮定し，それぞれ bx_i，dx_i とおけば

$$x_{i+1}-x_i=bx_i-dx_i$$

となる。この式を変形し，$c=1+b-d$ とおけば

$$x_{i+1}=cx_i \quad (i=0,\ 1,\ 2,\ \cdots\cdots)$$

を得る。この解は

$$x_i=c^i x_0 \qquad \cdots\cdots(1)$$

である。実際には $c=\dfrac{x_1}{x_0}$ として，x_2，x_3，……を計算できる。文献(5)は England と Wales の1880～1887年の人口について計算し，実際の値と推定値との差を求めているが，その誤差は年をおうにつれて大きくなっている。

筆者は，わが国の昭和30～50年の人口について試みたが，これについても同じ傾向がみられた。なお，わが国の人口は5年おきの値しか入手できなかったので，計算結果は5年おきに出力してある（第1図)。この計算をコンピュータで行なう場合のプログラムも BASIC を用いて示しておいた〔BASIC およびその一般教育における利用については，文献(12)参照〕。

第1図

```
POPULATION CHANGE 1
  (ﾀﾝｲ 1000ﾆﾝ)

ﾈﾝ          ｽｲﾃｲ          ｼﾞﾂｻｲ        ｺﾞｻ
 30          89275         89275        0
 35          93418.93      93419        -6.25E-2
 40          97755.25      98275        -519.75
 45          102292.8      103720       -1427.187
 50          107041        111500       -4459


PANAFACOM     UMOS/T  E003  BASIC    V01-L01

     PROGRAM NAME =  POP1

     010 PRINT #1,'POPULATION CHANGE 1'
     015 PRINT #1,'  (ﾀﾝｲ 1000ﾆﾝ)'
     016 DIM A(5)
     020 MATREAD A
     030 PRINT #1,' '
     050 C=A(2)/A(1)
     070 X=A(1)
     080 PRINT #1,'ﾈﾝ','ｽｲﾃｲ','ｼﾞﾂｻｲ','ｺﾞｻ'
     090 FOR I=1 TO 5
     100 PRINT #1,25+5*I,X,A(I),X-A(I)
     120 X=X*C
     130 NEXT I
     140 PRINT #1,' '
     990 DATA 89275,93419,98275,103720,111500
     999 END
```

こんどは，ある年 t における0～14歳，15～44歳，45歳以上の人口をそれぞれ $n_1(t)$, $n_2(t)$, $n_3(t)$ とし，人口ベクトル

$$N(t)=\begin{bmatrix} n_1(t) \\ n_2(t) \\ n_3(t) \end{bmatrix}$$

を定義する。ここで，出生数は $n_2(t)$ のみに比例するとし $b_2n_2(t)$ とおき，死亡数は，上の各区分ごとに，その人口に比例するとし，比例定数をそれぞれ d_1, d_2, d_3 とする。そうすれば0～14歳の区分の1年後の人口は，その1年間の出生数と，その1年間に生き残ったもので15歳にならなかったものの数との和に等しいとして，だいたい

$$n_1(t+1)=b_2n_2(t)+\frac{14}{15}(1-d_1)n_1(t)$$

が成り立つ。同様に

$$n_2(t+1)=\frac{1}{15}(1-d_1)n_1(t)+\frac{29}{30}(1-d_2)n_2(t)$$

$$n_3(t+1)=\frac{1}{30}(1-d_2)n_2(t)+(1-d_3)n_3(t)$$

行列を用いて，これらの式をかけば

$$N(t+1)=\begin{bmatrix}\frac{14}{15}(1-d_1) & b_2 & 0\\ \frac{1}{15}(1-d_1) & \frac{29}{30}(1-d_2) & 0\\ 0 & \frac{1}{30}(1-d_2) & 1-d_3\end{bmatrix}N(t)$$

となる。係数の 3 × 3 行列を A とおけば

$$N(t+1)=A\cdot N(t)$$

とかけて，この解は

$$N(i)=A^i\cdot N(0)\quad (i=0,\ 1,\ 2,\ \cdots\cdots) \qquad \cdots\cdots(2)$$

となる。文献(5)では

$$b_2=0.075$$

$$d_1=0.20,\ d_2=0.008,\ d_3=0.062$$

として，England と Wales の 1881～1887 年の人口の推定値を計算し，(1) よりも誤差が小さくなることを示している。

次に，この種のモデルをわが国の人口に適用してみよう。わが国においては，年齢の 3 区分として，0～14歳，15～64歳，65歳以上という資料が入手できた。この区分は労働人口を把握するのによいかもしれないが，上のモデルにとって適当であるかどうか疑問があろう。この人口をそれぞれ $n_1(t)$, $n_2(t)$, $n_3(t)$ とすれば，行列 A は

$$\begin{bmatrix}\frac{14}{15}(1-d_1) & b_2 & 0\\ \frac{1}{15}(1-d_1) & \frac{49}{50}(1-d_2) & 0\\ 0 & \frac{1}{50}(1-d_3) & 1-d_3\end{bmatrix}$$

第2図

```
POPULATION CHANGE 2
 (ﾀﾝｲ1000ﾆﾝ)

B2=          3E-2             D1=          4E-3
D2=          4E-3             D3=          5.999999E-2


ﾈﾝ           ｽｲﾃｲ             ｼﾞｯｻｲ        ｺﾞｻ
 30           89274            89275       -1
 35           94087.43         93419        668.4375
 40           98256.75         98275       -18.25
 45           101890.5         103720      -1829.437
 50           105097           111940      -6842.937
```

```
PROGRAM NAME =  POP2R

005 PRINT #1,'   POPULATION CHANGE 2'
006 PRINT #1,'    (ﾀﾝｲ1000ﾆﾝ)'
010 DIM A(5),B(3,3),N1(3),N2(3)
015 PRINT #1,' '
020 READ B2,D1,D2,D3
025 PRINT #1,'B2=',B2,'D1=',D1
026 PRINT #1,'D2=',D2,'D3=';D3
027 PRINT #1,' '
030 MAT B=ZER
040 B(1,1)=.933*(1-D1)
050 B(1,2)=B2
060 B(2,1)=.067*(1-D1)
070 B(2,2)=.98 *(1-D2)
080 B(3,2)=.02 *(1-D2)
090 B(3,3)=1-D3
100 MAT READ A,N1
115 J=1
120 PRINT #1,' '
130 PRINT #1,'ﾈﾝ','ｽｲﾃｲ','ｼﾞｯｻｲ','ｺﾞｻ'
140 FOR Y=1 TO 25
150 X=N1(1)+N1(2)+N1(3)
160 MAT N2=B*N1
170 MAT N1=N2
180 IF MOD(Y-1,5)<>0 GOTO 190
185 PRINT #1,Y+29,X,A(J),X-A(J)
186 J=J+1
190 NEXT Y
850 DATA .03,.004,.004,.06
880 DATA 89275,93419,98275,103720,111940
900 DATA 29798,54729,4747
980 PRINT #1,' '
999 END
```

となる。文献(10)と(11)によって，昭和30年における，総人口，出生数，3区分別の人口および死亡数にもとづく諸定数を

$$b_2=0.03$$

$$d_1=0.004,\ d_2=0.004,\ d_3=0.06$$

のように定めて計算した結果を第2図に示した。20年間にわたるものなので誤差は小さいとはいえないが，35年から40年にかけて誤差の符号が変わる点に注目したい。

例によって，コンピュータを用いて，この計算をするときのプログラムを BASIC で書いたものをつけておいた。

昭和30年から45年にかけての b_2, d_1, d_2, d_3 の値は次のようになっている。

年	b_2	d_1	d_2	d_3
30	0.032	0.0041	0.0048	0.064
35	0.027	0.0028	0.0043	0.069
40	0.027	0.0020	0.0036	0.066
45	0.027	0.0016	0.0033	0.060

b_2, d_1, d_2 の値は減少する傾向にあるが，d_3 は35年に増加しそれ以後は減少している。

このモデルは，年齢3区分別を，最初の例と同じにしてやってみたいと思うが，現在のところ，資料がないためにできないのが残念である。

4. 人口推定モデル（その2）

現象を数学的に表現するのに，離散量として扱う場合と，連続量的に扱う場合とがあって，前者のような扱いでは，前項の例でみたように差分（定差）方程式の理論や，行列の理論が役に立つ場合が多い。

これに対し，連続量として扱う場合は，微分方程式の理論，一般には，微分積分学が有効な道具となることが多い。幸いにして，文献(6)に，そのような立場からの人口推定モデルがのっている。この項では，これについ

て述べよう。

まず， T. Malthus の1978年の論文にもとづく，「Malthus の人口モデル」が紹介されている。時間 t におけるある国の人口を $N(t)$ で表し，出生数も死亡数も，人口と時間間隔 $\varDelta t$ に比例すると仮定すれば，α, β は定数として

$$\text{出生数} = \alpha\, N(t)\varDelta t$$

$$\text{死亡数} = \beta\, N(t)\varDelta t$$

$\varDelta t$ の間における $N(t)$ の変化を $\varDelta N$ とすれば

$$\begin{aligned}\varDelta N &= \alpha N(t)\varDelta t - \beta N(t)\varDelta t\\ &= (\alpha-\beta)N(t)\varDelta t\end{aligned}$$

ここで $\gamma=\alpha-\beta$ とおけば

$$\varDelta N = \gamma N(t)\varDelta t$$

すなわち
$$\frac{\varDelta N}{\varDelta t} = \gamma N(t)$$

$\varDelta t \to 0$ として

$$\frac{dN}{dt} = \gamma N$$

が得られ，$N(0)=N_0$ とすれば，この微分方程式の解は

$$N = N_0 e^{\gamma t} \qquad \cdots\cdots(3)$$

となる。Nは，$\gamma>0$ ならば増加，$\gamma<0$ ならば減少，$\gamma=0$ ならば一定である。

γは，$N(0)$ と $N(1)$ から

$$\gamma = \log\frac{N(1)}{N(0)}$$

によって求められる。文献(6)では，アメリカ合衆国における1790～1930年の人口について，(3)による推定値と，実際の値とを比較している。誤差は，年ごとに大きくなり，1930年には，それが，実際の値の約1.5倍にまで達している。

(3)をわが国の場合にあてはめて，Nを計算してみると第3図のようになる。

なお，手もとの資料には，5年おきの人口しかないので，γ に5年間の

第3図

```
POPULATION CHANGE 3
 (ﾀﾝｲ 1000ﾆﾝ)

ﾈﾝ          ｽｲﾃｲ          ｼﾞﾂｻｲ        ｺﾞｻ
 30          89276         89276        0
 35          93418.81      93419       -0.1875
 40          97754.06      98275       -520.9375
 45          102290.3      103720      -1429.687
 50          107037.2      111500      -4462.75



PANAFACOM     UMOS/T  E003  BASIC     V01-L01

     PROGRAM NAME =  POP3

     005 PRINT #1,'POPULATION CHANGE 3'
     006 PRINT #1,' (ﾀﾝｲ 1000ﾆﾝ)'
     010 DIM A(5)
     015 PRINT #1,' '
     100 MATREAD A
     120   G=LOG(A(2)/A(1))/5
     130 PRINT #1,'ﾈﾝ','ｽｲﾃｲ','ｼﾞﾂｻｲ','ｺﾞｻ'
     140 N0=A(1)
     150 FOR T=1 TO 21 STEP 5
     160 Y=T-1
     170 X=N0*EXP(G*Y)
     180 I=INT(Y/5)+1
     190 PRINT #1,30+Y,X,A(I),X-A(I)
     200 NEXT T
     990 DATA 89276,93419,98275,103720,111500
     998 PRINT #1,' '
     999 END
```

平均値をあてはめた。すなわち

$$\gamma=\frac{1}{5}\log\frac{N(4)}{N(0)}$$

としてある。(3)による推定値は，(1)による推定値に近いことに注意しよう。

文献(6)では，このあとに「Verhulst の改良モデル」というのが紹介してある。これは，1837年の論文によるもので，上の微分方程式の右辺に，crowding factor $-\lambda N^2$ をつけ加えて

$$\frac{dN}{dt}=\gamma N-\lambda N^2$$

とした。ここに，λ も定数，Nが大きくなれば，$-\lambda N^2$ は人口の増加に対

してブレーキをかける。物理学における〈抵抗〉にヒントを得たものであろう。

この微分方程式は，いわゆる変数分離法によって，次のように解ける。

$$\int \frac{dN}{N(1-\lambda N/\gamma)} = \int \gamma dt$$

$$\therefore \int\left(\frac{1}{N} - \frac{\lambda/\gamma}{1-\lambda N/\gamma}\right)dN = \gamma t + \mathrm{K}$$

ただし，Kは任意定数

両辺の積分を計算して

$$\log N - \log\left(1-\frac{\lambda N}{\gamma}\right) = \gamma t + \mathrm{K}$$

$N(0)=N_0$ とすれば

$$\mathrm{K} = \log \frac{N_0}{1-\lambda N_0/\gamma}$$

$$\therefore \frac{N}{1-\lambda N/\gamma} = \frac{N_0 e^{\gamma t}}{1-\lambda N_0/\gamma}$$

したがって

$$N(t) = \frac{N_0 \gamma e^{\gamma t}}{\gamma + \lambda N_0 (e^{\gamma t} - 1)} \qquad \cdots\cdots(4)$$

を得る。ここで，$t \to \infty$ とすれば $N(t) \to \frac{\gamma}{\lambda}$ となる。アメリカ合衆国の1800～1930年の人口を(4)で推定する場合は

$$N_0 = 3.9 \times 10^6,\ \gamma = 0.3134,\ \lambda = 1.5337 \times 10^{-9}$$

とすれば，実際の値とよく合う由である。

これを，わが国の例にあてはめようとすれば，N_0 と γ の値は，(3)を用いた場合と同じだから問題ないが，γ の値をどのように定めたらよいか問題になる。これは，数学的には実験データに合うようにパラメータの値を決める問題であり，一般にはなかなかめんどうなものであるが，コンピュータを利用すれば比較的楽に，いろいろな λ の値を試みてみることができる。第4図には，そのような形のプログラムを挙げておいた。

これによって，いろいろな λ（プログラム中ではL）の値を試みてみると，わが国の例では，$\lambda \geqq 0$ のなかでは，$\lambda = 0$ のとき（すなわち，(3)の場合）に誤差の絶対値の和が最小になる。そして，$\lambda = -1 \times 10^{-8}$（プログラ

第4図

```
POPULATION CHANGE 4
  (ﾀﾝｲ 1000ﾆﾝ)

L=         -1E-9

ﾈﾝ          ｽｲﾃｲ           ｼﾞﾂｻｲ         ｺﾞｻ
30          89275          89275         0
35          93461.56       93419         42.5625
40          97846.68       98275         -428.3125
45          102439.8       103720        -1280.187
50          107251         111500        -4249


PANAFACOM      UMOS/T  E003  BASIC     V01-L01

    PROGRAM NAME =  POP4

    010 PRINT #1,'POPULATION CHANGE 4'
    015 PRINT #1,'  (ﾀﾝｲ 1000ﾆﾝ)'
    020 PRINT #1,' '
    025 DIM A(5)
    030 MATREAD A
    040  G=LOG(A(2)/A(1))/5
    050  N0=A(1)
    060 PRINT 'L=';
    070 INPUT L
    080 IF L=999 GOTO 980
    090 PRINT #1,'L=',L
    095 PRINT #1,' '
    100 PRINT #1,'ﾈﾝ','ｽｲﾃｲ','ｼﾞﾂｻｲ','ｺﾞｻ'
    110 FOR T=1 TO 21 STEP 5
    120 N=N0*EXP(G*(T-1))
    130 X=G*N/(G+L*N-L*N0)
    140 I=INT(T/5)+1
    150 PRINT #1,29+T,X,A(I),X-A(I)
    160 NEXT T
    165 PRINT #1,' '
    170 GOTO 60
    980 PRINT #1,' '
    990 DATA 89275,93419,98275,103720,111500
    999 END
```

ム中では−1E−8）のとき，さらによくあてはまる。

ただし，このような曲線のあてはめが，実際上，どんな意味をもつか明確でない。あるいは，このモデルの限界を示しているのかもしれない。

5. おわりに

ここで挙げた人口推定モデルは，数学的な面からみれば，高校から大学初年級にかけての練習問題にふさわしい。しかし，これを現実の現象に結びつけていくことになると決して易しくないように見える。

また，文献(5)，(6)にはなかったが，これらの数学モデルを扱うに当たってコンピュータを利用するのはきわめて効果的であろう。むしろ，「計算機実習」の中級以上の題材として適しているかもしれない。さらにくふうしてみたいと考えている。最後に，いろいろとお世話になった本学の加藤寿延教授，藤井良治助教授，前記研究班の諸氏，特に，国立教育研究所の島田茂氏，大阪教育大学の三輪辰郎氏に感謝の意を表したい。なお，本研究は科学研究費補助金による総合研究の一部をなすものである。

(注) コンピュータのプログラム形式について本文中でふれる余裕がなかったので，若干の注意をつけ加えておく。

(1) 第1～4図のプログラムは，ラインプリンターに計算結果を出力するために，プリント命令に，ラインプリンターの機番 ♯1 が入っているが，ふつうの BASIC ではこれを省略してよい。

(2) MAT のついた命令は，すべて行列計算関係のものである。

(3) 推定値はすべて四捨五入によって整数値にすべきであるが，本学のコンピュータでは，この種の計算のプログラムが複雑になるので，小数点以下も出力しておいた。

文 献

1) J. G. Kemeny & J. L. Snell: Mathematical modeles in Social Sciences, MIT Press (1962)
2) J. G. Andrews: Mathematical modelling, Butterworth (1976)
3) H. P. Williams: Model building in Mathematical Programming, John Wiley & sons (1978)
4) J. Lighthill: Newer uses of Mathematics, Penguin Books (1978)
5) D. N. Burghes & M. S. Borrie: Mathematical Modelling*

6）D. N. Burghes & M. S. Borrie: Modelling with Calculus*
7）近藤次郎：数学モデル，丸善（1976）
8）赤摂也他：数理科学の諸問題，筑摩書房（1971）
9）黒田俊夫：日本人の寿命，日経新書（1978）
10）昭和50年人口動態統計　第1巻，厚生省
11）人口ハンドブック（昭和50年度），人口問題研究会（非売品）
12）植竹恒男：一般教育としての情報処理教育について，亜細亜大学教養部紀要第16号（1977）

＊　5），6）については，本文参照。